
· [bookmark: RTF370039003700370037003a00]Messages
· Overview
The messaging feature is a centralized and flexible mechanism used to write text messages to various destinations, such as log files, or the display, waveforms or databases. It lets a developer easily insert formatted messages into code and provides the user with powerful and flexible controls to selectively enable or disable groups of messages.
The three most typical uses for messages are the following:
· Summaries—Writing summary information at the beginning or end of significant chunks of activity.
· Tracing—Writing detailed trace messages during the simulation upon interesting events.
· Debugging—Writing detailed debug messages during the run to help the user or developer debug unexplained behaviors.
Messages are different from plain out() and outf() calls (see 29.7); they have an optional standard-format prefix and their actions can be disabled or redirected. Messages are also different from dut_error() calls (see 17.2.2); they do not signify failure, increment error counters, or increment warning counters.
· Regular Message model
Regular messages are used to indicate “atomic” events: the information contained by the message does not span over time.
Upon execution, the message action creates a message and sends it to a handling unit ( it’s immediate enclosing unit) for further handling. one or more message loggers that have been configured to be a handling logger for this message. Each handling loggerunit can be configured to filter messages in various ways, format the enabled messages in various ways (adding the time, name of the unit, etc.), and send them to various destinations (files and the screen).
The message loggers are instantiated by the programmer in the unit hierarchy and then configured using constraints (see 25.5) or method calls (see 25.6).
There is a predefined message logger instantiated in every environment under sys called logger.
· message and messagef 
	Purpose
	Create a (formatted) message and send it to handling unita message logger

	Category
	Action

	Syntax
	message ([tag], verbosity, exp, ...) [action_block]
messagef ([tag], verbosity, format_exp, [exp, ...]) [action_block]

	Parameters
	tag
	A constant of type message_tag, either NORMAL or a user-defined tag (see 25.3.1).

	
	verbosity
	A constant of type message_verbosity: one of NONE, LOW, MEDIUM, HIGH, or FULL (see 25.3.2).

	
	exp
	Value(s) to write.

	
	action_block
	A block of actions to perform, the output of which is taken as part of the message output.

	
	format_exp
	For messagef(), an outf()-style format string for the output.


· 
When a message() or messagef() action is executed, the following happens:
· If there are no handling loggers for the action (see 25.4.2), then the action is skipped.
· If there are handling loggers for the action, tThe action creates a message (consisting of a list of strings, plus related information) which is further processed by handling unit in order to drop it or send it to it’s assigned destination and sends it to all of the handling loggers. Those loggers format the message and send it to the screen and/or log files.
· For message(), the first string in the message is created by appending all of the expressions, like out() does.
· For messagef(), the first string is created using the format-exp, similar to outf().
· messagef() does not automatically add a newline (\n) to the message string. Therefore, if the optional action-block requires a newline to be written before it is executed, terminate the format-exp using \n.
· If the fully composed message string—including that portion written by the optional action-block—is not terminated by a newline, a newline is appended. messagef() also allows appending of the action-block output to the messagef() header output.
· If an action-block exists, it gets executed. This typically contains further output-producing actions, calls to reporting methods, etc. The output of all of those is added, as a list of string, to the message.
Message code shall not modify the flow of the simulation in any way. Time-consuming operations in message headers or action blocks are strictly disallowed.
Syntax examples:
message(HIGH, "Master ", me, " has received ", the_packet) {
    write the_packet
};
    -- Output this message and write the packet, at verbosity HIGH.

message(VR_XBUS_FILE, MEDIUM, "Packet ", num, " sent: ", data)
    -- Output this message at verbosity MEDIUM.
    -- Use VR_XBUS_FILE as the message-tag.
· [bookmark: RTF370037003500330037003a00]Tag
Both message() and messagef() have an optional first parameter of type message_tag, which is initially defined as:
type message_tag : [NORMAL]
This can be extended, e.g.,
extend message_tag : [VR_XBUS_PACKET]
If a tag is not specified [i.e., the first parameter of message() or messagef() is a legal value for verbosity], then the value NORMAL is prepended. Thus, the following two lines are the same:
message(        MEDIUM, "Packet done: ", packet);
message(NORMAL, MEDIUM, "Packet done: ", packet)
Message tags are used for associating specific message actions with a a message logger class of actions. This gives you more flexibility when it comes to determine the behavior of the message actions.  (see 25.4).
· [bookmark: RTF390039003300340033003a00]Verbosity
The verbosity parameter can be set to NONE, LOW, MEDIUM, HIGH, or FULL (from lowest to highest). Since a lower verbosity setting means fewer messages are shown, important messages should be assigned a lower verbosity parameter value.
Table 46 shows the recommended usage of verbosity. Each level can assume that all lower levels are also writing (thus, there is no need to repeat them). 
	· [bookmark: RTF310031003800350032003a00]Verbosity levels 

	Level
	Recommended use
	Examples

	NONE
	Critical messages (this level cannot be disabled).
	"WARNING: Running in reduced mode"

	LOW
	Messages that happen once per run or once per reset.
	"Master M3 was instantiated"
"Device D6 got out of reset"

	MEDIUM
	Short messages that happen once per data item or sequence.
	"Packet-@36 was sent to port 7"
"A write request to pci bus 2 with address=0xf2223, data=0x48883"

	HIGH
	More detailed per-data-item information, including:
· Actual value of the packet
· Sub-transaction details
	"Full details for packet-@36: len=5 kind=small ..."

	FULL
	Anything else, including writing by using specific methods (just to follow the algorithm of that method).
	



· [bookmark: RTF370035003400380039003a00]Nested message actions
A message action may be designated as a leader by using the set_leader() method (see 25.6.1.7). A leader message action is responsible for all message actions inside its lexical scope or in methods called from it. All nested message actions are handled by the leader’s loggers (and only by them). The nested messages’ text is formatted sequentially according the leader’s rules, as if their output code resides directly inside the leader’s lexical scope. This is true even if some of the nested message actions are themselves designated as leaders—the outermost leader overrides the rest.
For sample usage of nested message actions, see 25.7.2 and 25.7.3.
· [bookmark: RTF330035003500340037003a00]Structured Debug Message (SDM) Actions
Structured debug messages(SDM) are used to log transactions or sequences of events that span over time (eg whole bus transactions) to a destination: log file, display, database. While the regular message actions are used to log string messages, the SDM also samples data objects given as arguments which can further be used for data flow analysis (e.g. using a waveform viewer). The sampling of the arguments of SDM actions is related to transaction recording.
The e language provides the following methods for defining structured debug message actions:
●    msg_started()—Marks the beginning of a sequence of events that you can log as a transaction.
●    msg_ended()—Marks the end of a sequence of events that you can log as a transaction.

●    msg_transformed()—Marks a data transformation. You can use msg_transformed() to link transactions that have been created by structured debug messages.
●    msg_changed()—Marks a change to an object, such as a state variable changing from transmit to idle. This message action marks a one-time event. It s logged as a transaction with a single attribute, the result of the state expression.
●    msg_info()—Reports any other kind of significant event in the environment. Events reported by
msg_info() cannot be logged as transactions.


SDM actions have the generic format described in table Table 47. The specifics for each of the SDMs are described in separate/dedicated sections.


Table 47 – SDM Action Generic Format
	Syntax
	msg_<sdm specifier>([tag,]verbosity, msg-id,< sdm specific arguments>) [{action-block}]

	
	sdm specifier
	It can be one of : started (see 25.4.1), ended(see 25.4.2), transformed(see 25.4.3), changed(see 25.4.4), info(see 25.4.5).
It determines two aspects:
· List of arguments
· Message instance parameters that can be accessed within action-block scope

	Parameters
	tag
	A constant of type message_tag, either NORMAL or a user-defined tag (see 25.3.1).

	
	verbosity
	A constant of type message_verbosity: one of NONE, LOW, MEDIUM, HIGH, or FULL (see 25.3.2).

	
	msg-id
	The message ID.
A string expression that identifies the specific occurrence 
reported by the message (i.e. message ID uniquely 
identifies a transaction stream). 
When a literal string is provided (as opposed to a string expression 
that is computed at runtime), the text can be used for static message 
filtering.

	
	SDM specific arguments
	These are determined by each SDM action and they are described in the sections 25.4.1-25.4.5 for each of the SDM actions.
Usually these are objects that will be sampled for later analysis purposes.


	
	Action-block
	A list of zero or more “actions” separated by semicolons and enclosed in curly braces. Syntax: {action;...} The action-block is executed once if the message action is not discarded due to filtering.
The primary use of this action-block is to initialize configurable message instance parameters (described below) with values that you assign to the fields of the pseudo variable it. Example:
msg_started(HIGH," monitoring transfer",cur_trans) {
        it.parent = cur_burst;
};
However, you can use this action-block like any action-block in e.

The configurable message instance parameters in the action block are fields of the message handler struct for the message action. You can set these fields by referring to them as it.field. The following configurable message instance parameters are common to all SDM actions; the ones specific to each SDM are described in the specific sections.
	scope
	Identifies the unit context where the action occurs. This can be used, for example, to hide the actual unit and use an enclosing unit as the message scope.
The default is me if the message is issued within the scope of a unit type. Otherwise, the default is the value returned from it.

	body_text
	Defines a text string, or a list of text strings separated by commas, to be displayed with the message. Use this field to override the message’s default text, which is a hyperlink to the logged transaction.






The developer can configure transaction recording process and specify what to sample from a transaction object and when to sample it. For more information see chapters: recording_config API, any_unit messaging API, any_struct messaging API. 

25.4.1 msg_started()
	Purpose
	Reports the start of a transaction.

	Category
	Action

	Syntax
	msg_started([tag,]verbosity, msg-id, data-item) [{action-block}]


	Parameters
	tag, verbosity, msg-id
	See Table 47 SDM Action Generic Format

	
	data-item
	The struct that contains the data that is being processed.


	
	Action-block
	See Action-Block description in Table 47 SDM Action Generic Format
msg_started can also accept the following configurable message instance parameter, besides the ones presented in Table 47 SDM Action Generic Format

	parent
	Identifies the higher-level (parent) transaction containing the current transaction.
If specified, the struct assigned to the .parent field becomes the parent transaction, and the data item of the current transaction becomes the child transaction.
This can be useful, for example, for showing to which burst a set of packets belongs. (Transactions are usually used to model “packets,” and bursts are the children of “transfers”; thus, the parent attribute for each packet points to the “burst” message, and parent of the burst points to a transfer.)







25.4.2 msg_ended()
	Purpose
	Reports the end of a transaction. 

	Category
	Action

	Syntax
	msg_ended([tag,]verbosity, msg-id, data-item) [{action-block}]

	Parameters
	tag, verbosity, msg-id
	See Table 47 SDM Action Generic Format

	
	data-item
	The struct that contains the data that is being processed.


	
	Action-block
	See Action-Block description in Table 47 SDM Action Generic Format
msg_ended can also accept the following configurable message instance parameters, besides the ones presented in Table 47 SDM Action Generic Format

	parent
	Identifies the higher-level (parent) transaction containing the current transaction.
If specified, the struct assigned to the .parent field becomes the parent transaction, and the data item of the current transaction becomes the child transaction.Use this optional parameter only when there is no corresponding msg_started() action. In this case, msg_ended() should also specify the transaction start_time.
This parameter can be useful, for example, for showing to which burst a set of packets belongs. (Transactions are usually used to model “packets,” and bursts are the children of “transfers”; thus, the parent attribute for each packet points to the “burst” message, and parent of the burst points to a transfer.)

	start_time
	The time at which this transaction started (by default UNDEF, which indicates that the starting of the transaction was already reported by a msg_started() action). Must be a value of type time.







Reports the end of each transaction that you want to track. Unless you specify the sample points with the recording configuration API, data is sampled as follows:

●    When a msg_ended() action has a corresponding msg_started() action, data is sampled at both the beginning and end of the transaction.


●    When a msg_ended() action has no corresponding msg_started() action, you can specify the start time in the body of the msg_ended() action. In this case, data is sampled at the end of the transaction.
●    When a msg_ended() action has no corresponding msg_started() action, and no start time is set in the action body, a 0-time transaction is created, and data is sampled at the end of the transaction.
25.4.3 msg_transformed()
	Purpose
	Reports the transformation of an existing data item or items, or the outcome of a relationship between data items.

	Category
	Action

	Syntax
	msg_transformed([tag,] verbosity, msg-id, from-item, to-item) [{action-block}]

	Parameters
	tag, verbosity, msg-id
	See Table 47 SDM Action Generic Format

	
	from-item
	The struct containing the data that is being processed. 

	
	to-item
	The struct containing the data after transformation.


	
	Action-block
	See Action-Block description in Table 47 SDM Action Generic Format



25.4.4 msg_changed()
	Purpose
	Reports a significant state change taking place in this scope.

	Category
	Action

	Syntax
	msg_changed([tag,] verbosity, msg-id, new-state-exp) [{action-block}]

	Parameters
	tag, verbosity, msg-id
	See Table 47 SDM Action Generic Format

	
	new-state-exp
	A text string describing the new state this unit assumes. 

	
	Action-block
	See Action-Block description in Table 47 SDM Action Generic Format



25.4.5 msg_info() 
	Purpose
	Reports a significant event in the environment, that occurs at a certain point in time, possibly related to the provided data items or items, and which is not applicable to the other kinds of structured debug messages.

	Category
	Action

	Syntax
	msg_info([tag,]verbosity, msg-id, data-item1 [, data-item2]) [{action-block}]

	Parameters
	tag, verbosity, msg-id
	See Table 47 SDM Action Generic Format

	
	data-item-1,
data-item-2
	References to data items involved in the reported event. You can 
specify up to two data items (the second one is optional).

	
	Action-block
	See Action-Block description in Table 47 SDM Action Generic Format
msg_ended can also accept the following configurable message instance parameters, besides the ones presented in Table 47 SDM Action Generic Format

	data_item
	The struct containing  the data that is being processed.






[bookmark: _GoBack]
Message loggers
Overview
message_logger is a predefined unit type that is used to manage the output from message actions: filtering it, formatting it, or sending it to one or more destinations. Message loggers are defined programmatically and attached as fields to various units in the unit hierarchy, e.g.,
extend vr_xbus_env_u {
    logger : message_logger is instance
}
The following can be specified (via methods or constraints) for each message logger:
Which subset of the message actions the logger examines. By default, each logger uses a verbosity of NONE and its tag list is empty. This passes the control to the predefined logger sys.logger until the (new) logger is explicitly enabled.
Which subset of the unit instances the logger examines. By default, this is the tree starting at the unit to which the logger is attached, e.g., each vr_xbus_env_u.logger looks at the unit sub-tree under the corresponding vr_xbus_env_u unit.
Which destinations (files or screen) to use for receiving output.
What format to use.
Loggers, like other units, are generated at elaboration time before the simulation run begins.
[bookmark: RTF340038003300360031003a00]Handling messages
During execution, messages are processed by loggers in the following manner:
Upon execution of a message action, the set of loggers relevant to that message is determined. The relevant messages are those for which both the following conditions hold:
They are associated with the object’s origin unit—the unit instance returned by any_struct.get_unit() (see 7.5.1).
The current setting applies to them according to its verbosity, tag, and other criteria (see 25.6.1).
Then, for each of the relevant loggers:
accept_message() is called (see 25.6.2); if it returns FALSE (due to explicit user refinement), the logger is skipped.
format_message() is called (see 25.6.2) to retrieve the final message format.
The message in its final format is written to each of the destinations assigned to the logger (the screen and/or any log files).
Each message may be sent to a given destination at most once. If 
more than one logger is instantiated at the same point in the unit hierarchy;
two or more loggers handle the same message and are assigned the same destination; and
more than one candidate logger still remains after evaluating the other criteria (verbosity, tag, accept_message(), etc.).
Then the message shall go to all distinct destinations defined in the candidate loggers, and the format used when more than one candidate logger specifies the same message shall be implementation-dependent.
[bookmark: RTF330033003400370037003a00]Configuring message loggers with constraints
Loggers may be configured by using constraints or through a messaging interface (see 25.6). Constraints are used during pre-run generation to set the fields of the logger. Then, during post_generate() (see 10.5.3), the logger fields are used to configure the logger. At post-generation, the logger becomes attached to the unit [specifically, the unit computed by logger-instance.get_unit()]. The post_generate() method of a logger uses the generated values of the fields to configure the logger via the procedural interface (PI).
Table 47 shows the constrainable fields of the unit logger, along with the built-in soft constraints that determine their default values. 
	[bookmark: RTF390033003700380031003a00]Logger constrainable fields 

	Field
	Description

	tags : list of message_tag;
    keep soft tags == {}
	The message tags for selecting the actions for this logger.

	verbosity : message_verbosity;
    keep soft verbosity == NONE
	The verbosity for selecting the actions for the logger.

	modules : string;
    keep soft modules == "*"
	The modules wild card for selecting the actions for the logger.

	string_pattern : string;
    keep soft string_pattern == "..."
	The pattern to match against the string in the message action.

	to_file : string;
    keep soft to_file == ""
	The file to which the logger writes (or none, if the setting is " ").

	to_screen : bool;
    keep soft to_screen == TRUE
	When set to TRUE, the logger also writes any output to the screen.



The following considerations also apply:
Only one file may be associated with a logger via constraints. However, multiple files are allowed when using the messaging procedural interface (see 25.6).
By default, loggers ignore all tags and have a verbosity of NONE. The one exception is sys.logger, where the tag list defaults to {NORMAL} and the verbosity defaults to LOW. Thus, all messages are controlled by sys.logger.
If the verbosity is constrained to a value other than NONE and the tag list is unconstrained, the tag list defaults to {NORMAL}.
The name of the file written by the logger is appended with the extension .elog if the value of the field to_file specified by the user does not include an extension. In this context, an extension is a period (.), followed by one or more characters in the portable filename character set.
[bookmark: RTF390039003200350038003a00]Messaging procedural interface (PI)
The following methods of struct type message_logger can be used to set its configuration before or during execution, refine its filtering or formatting rules, and query for details of the currently processed message action. These methods may be used at runtime to override the default (see 25.5) or user-defined settings specified by constraints.
[bookmark: RTF320039003000360039003a00]Methods for setting configuration
The following methods of struct type message_logger are used to set its configuration.
[bookmark: RTF370033003300320030003a00]set_actions 
	Purpose
	Add, remove, or replace the specified actions for the logger

	Category
	Method

	Syntax
	set_actions (verbosity, tags, modules, text, op)

	Parameters
	verbosity
	A constant of type message_verbosity: one of NONE, LOW, MEDIUM, HIGH, or FULL (see 25.3.2). This matches the message actions whose verbosity is between LOW and verbosity (when op = add or replace) or between verbosity and FULL (when op = remove).

	
	tags
	Matches the message actions whose tags (see 25.3.1) are those specified by the list (of type message_tag).

	
	modules
	Matches the message actions residing in the module(s) that match this string expression.

	
	text
	Matches the message actions, of which one of the parameters is a literal string that matches the string expression.

	
	op
	One of add, replace, or remove. This determines whether the message actions matched by previous parameters are added to the logger, removed from it, or used to replace the currently assigned action.


This adds, removes, or replaces the specified actions for the logger.
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        logger.set_actions(FULL, {NORMAL}, "*", "...", replace)
    }
}
set_unit 
	Purpose
	Set the unit tree under a specified unit to be on or off for the message logger

	Category
	Method

	Syntax
	set_unit (root, status)

	Parameters
	root
	The root unit of the unit tree to add to or remove from the set of unit instances associated with the logger.

	
	status
	One of on or off; this determines whether the units are added or removed.



This adds (on) or removes (off) units from the logger.
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        logger.set_unit(testbench_01, off)
    }
}
set_format 
	Purpose
	Set the format for messages associated with the message logger

	Category
	Method

	Syntax
	set_format (format)

	Parameters
	format
	A value of type message_format. The predefined choices are short, long, or none; this can be extended by the user. none specifies no additions to the bare message text. Any styles implied by the other formats are implementation-dependent.



This sets the format mode for any messages associated with the logger.
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        logger.set_format(none)
    }
}
set_flush_frequency 
	Purpose
	Set the message flushing frequency of each message output file associated with the logger

	Category
	Method

	Syntax
	set_flush_frequency (num)

	Parameters
	num
	The maximum number of message actions executed before their content is flushed out.



This sets the maximum number of message actions allowed before flushing the message queue for the logger.
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        logger.set_flush_frequency(15)
    }
}
set_file 
	Purpose
	Add a file to or remove a file from the specified logger

	Category
	Method

	Syntax
	set_file (fname, status)

	Parameters
	fname
	The log file to add or remove.

	
	status
	One of on or off; this determines whether the files are added or removed.



This adds a file to the specified logger (or if off is specified, removes it from the logger). Each logger can write to any number of files, including writing to the screen (see 25.6.1.6).
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        var my_file : file;

        my_file = files.open("my.log", "w", "Log file");
        logger.set_file(my_file, on)
    }
}
[bookmark: RTF370037003400310039003a00]set_screen 
	Purpose
	Enable or disable the logger from writing to the screen

	Category
	Method

	Syntax
	set_screen (status)

	Parameters
	status
	Either on (writing to screen is enabled) or off (writing to screen is disabled).


This enables or disables the logger from writing to the screen. [Any messages sent to the screen are directed the same way as text from the out() pseudo-routine and the print action.]
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        if logger.to_file != "" then {
            logger.set_screen(off)
        }
    }
}
[bookmark: RTF330035003100330035003a00]set_leader 
	Purpose
	Set the specified messages’ actions as leaders or non-leaders.

	Category
	Method

	Syntax
	set_leader (verbosity, tags, modules, text, status)

	Parameters
	verbosity
	A constant of type message_verbosity: one of NONE, LOW, MEDIUM, HIGH, or FULL (see 25.3.2). This matches the message actions whose verbosity is between LOW and verbosity (when set_actions op = add or replace) or between verbosity and FULL (when set_actions op = remove). See also 25.6.1.1.

	
	tags
	Matches the message actions whose tags (see 25.3.1) are those specified by the list (of type message_tag).

	
	modules
	Matches the message actions residing in the module(s) that match this string expression.

	
	text
	Matches the message actions, of which one of the parameters is a literal string that matches the string expression.

	
	status
	One of on or off; this determines whether to set the specified messages’ actions as leaders (on) or non-leaders (off).


This sets the specified actions to be leaders (or if off is specified, to cease from being leaders); see also 25.3.3.
Syntax example:
extend vr_xbus_env_u {
    run() is also {
        logger.set_leader(LOW, {NORMAL}, "*", "...", on)
    }
}
ignore_tags 
	Purpose
	Ignore the specified tags

	Category
	Method

	Syntax
	ignore_tags (tags)

	Parameters
	tags
	Ignores the message actions whose tags are those specified by the list (of type message_tag).



This causes the logger to ignore messages of this tag type.
Syntax example:
extend message_tag : [VR_XBUS_PACKET];

extend vr_xbus_env_u {
    run() is also {
        logger.ignore_tags({VR_XBUS_PACKET})
    }
}
[bookmark: RTF340032003400340039003a00]Hook methods for refining message handling
The predefined methods shown in this subclause are automatically called during message handling. They can also be extended to override the default behavior.
[bookmark: RTF350036003200370037003a00]accept_message 
	Purpose
	Check if the current message is enabled

	Category
	Method

	Syntax
	accept_message():bool

	Parameters
	None
	

	Return value
	A Boolean value


This returns TRUE (the default) if the current message is enabled. When this method returns FALSE (see the following example), the current message is ignored by the logger and the logger destination is blocked for the current message.
Syntax example:
extend message_logger {
    accept_message(): bool is only {
        return get_tag() == normal
    }
}
[bookmark: RTF330039003500390036003a00]format_message 
	Purpose
	Format a message

	Category
	Method

	Syntax
	format_message():list of string

	Parameters
	None
	

	Return value
	The message


This returns the list of string, which will be sent as-is to the file or screen.
Syntax example:
extend message_logger {
    format_message(): list of string is only {
        var msg_list := get_message();
        ...
    }
}
[bookmark: RTF340036003400300034003a00]Query methods for getting message information
The methods shown in this subclause are available for use within the accept_message() (see 25.6.2.1) and format_message() methods (see 25.6.2.2). Query methods return information about the message action that was just executed.
get_format 
	Purpose
	Get a message’s format

	Category
	Method

	Syntax
	get_format():message_format

	Parameters
	None
	

	Return value
	The current message_format value for the logger. The built-in choices for message_format are short (the default), long, or none; the user is allowed to extend the type. 



This returns the message format of the current logger.
get_message 
	Purpose
	Get the current raw message

	Category
	Method

	Syntax
	get_message():list of string

	Parameters
	None
	

	Return value
	The message



This returns the current raw message as produced by the message action.
get_message_action_id 
	Purpose
	Get the message ID

	Category
	Method

	Syntax
	get_message_action_id():int

	Parameters
	None
	

	Return value
	The message ID



This returns a unique number identifying the message action.
get_tag 
	Purpose
	Get a message’s tag

	Category
	Method

	Syntax
	get_tag():message_tag

	Parameters
	None
	

	Return value
	A message-tag (see 25.3.1)



This returns the tag of the message action.
get_verbosity 
	Purpose
	Get a message’s verbosity

	Category
	Method

	Syntax
	get_verbosity():message_verbosity

	Parameters
	None
	

	Return value
	The message’s verbosity (see 25.3.2)



This returns the verbosity of the message action.
source_location 
	Purpose
	Get the message’s location

	Category
	Method

	Syntax
	source_location():string

	Parameters
	None
	

	Return value
	The source location



This returns the source location where the message occurred, e.g., "At line 12 in @foo".
source_method_name 
	Purpose
	Get the method name for the current message

	Category
	Method

	Syntax
	source_method_name():string

	Parameters
	None
	

	Return value
	The method’s name



This returns the name of the method where the message occurred, e.g., "foo".
source_struct 
	Purpose
	Get the message’s source struct

	Category
	Method

	Syntax
	source_struct():any_struct

	Parameters
	None
	

	Return value
	A struct



This returns the struct where the message occurred.
source_struct_name 
	Purpose
	Get a message’s struct type-name

	Category
	Method

	Syntax
	source_struct_name():string

	Parameters
	None
	

	Return value
	The struct’s type-name



This returns the name of the struct type where the message occurred, e.g., "packet".
Examples
Example 1
Add some message actions, and put a logger in the unit.
unit my_dsp {
    foo() is {
        ...
        message(LOW, "Starting transmission");
        -- LOW verbosity means this is an important message that
        -- will be shown even when verbosity is set to ’LOW’
        ...
        message(MEDIUM, "Sending packet ", pkt);
        -- MEDIUM verbosity means this is a less important message
        ...
    };

    logger: message_logger is instance
    -- Instantiate a message logger for this unit. When activated,
    -- it will handle all message actions executing in this unit
    -- or in any unit or struct under it.
}
Configure the logger via constraints.
extend sys {
    dsp : my_dsp is instance;
      keep dsp.logger.tags == {NORMAL};
      keep dsp.logger.verbosity == LOW;
      -- This logger will only look at important messages
      keep dsp.logger.to_file == "dsp_results.elog"
      -- Send it also to a file (it goes to the screen by default)
}
[bookmark: RTF320031003200340032003a00]Example 2
Confirm the writing order, assuming the following code:
message(NONE, "I’m A0") {
    out("A1");
    message(NONE, "I’m B0") {
        out("B1");
        message(NONE, "I’m C0") {
            out("C1")
        };
        out("B2")
    };
    out("A2")
}
If there is no leader message, then the output is written in the order that the message blocks finish:
I'm C0
C1
I'm B0
B1
B2
I'm A0
A1
A2
With "I'm A0" as the leader message, the order would be:
I'm A0
A1
I'm B0
B1
I'm C0
C1
B2
A2
[bookmark: RTF320032003000300031003a00]Example 3
This example illustrates the leader message concept. The message action "goo: I'm a message from A" is nested within the lexical scope of the message action "foo I'm U2". If neither of these messages is designated a leader (see 25.3.3 and 25.6.1.7), "goo: I'm a message from A" is handled by logger sys.u1.l1 and "foo I'm U2" by sys.u2.l2. However, if the outermost message action "foo I'm U2" is a leader, both messages are handled by sys.u2.l2, irrespective of the leader designation of the inner message action "goo: I'm a message from A".
extend sys {
    u1 : U1 is instance;
    u2 : U2 is instance;

    run() is also {
        u2.foo()
    }
};

unit U1 {
    l1 : message_logger is instance;
      keep l1.verbosity == FULL;
    a  : A
};

unit U2 {
    l2 : message_logger is instance;
      keep l2.verbosity == FULL;

    foo() is {
        message(NONE, "foo: I’m U2") {
            out("foo: before");
            sys.u1.a.goo();
            out("foo: after")
        }
    }
};

struct A {
    goo() is {
        message(NONE, "goo: I’m a message from A")
    }
}
If there is no leader message, then the output would be:
goo: I'm a message from A   -- handled by sys.u1.l1
foo: I'm U2                 -- handled by sys.u2.l2
foo: before                 -- handled by sys.u2.l2
foo: after                  -- handled by sys.u2.l2
To change the leader message, the following code can be added to the example:
extend sys {
    run() is first {
        u2.l2.set_leader(NONE, {NORMAL}, "*", "foo: I’m U2", on)
    }
}
With "foo: I'm U2" as the leader message, the order would be: 
foo: I'm U2                 -- handled by sys.u2.l2
foo: before                 -- handled by sys.u2.l2
goo: I'm a message from A   -- handled by sys.u2.l2
foo: after                  -- handled by sys.u2.l2

